# Maintainer: Jaroslav Lichtblau <svetlemodry@archlinux.org>
# Contributor: dibblethewrecker dibblethewrecker.at.jiwe.dot.org
# Contributor: William Rea <sillywilly@gmail.com>
# Contributor: Alexander Rødseth <rodseth@gmail.com>

# ALARM: Kevin Mihelich <kevin@archlinuxarm.org>
#  - --disable-inline to fix ARM FTBFS

pkgname=geos
pkgver=3.9.1
pkgrel=1
pkgdesc="C++ port of the Java Topology Suite"
arch=('x86_64')
url="https://trac.osgeo.org/geos/"
license=('LGPL')
depends=('gcc-libs' 'bash')
options=('!emptydirs')
changelog=$pkgname.changelog
source=(https://download.osgeo.org/$pkgname/$pkgname-$pkgver.tar.bz2)
sha256sums=('7e630507dcac9dc07565d249a26f06a15c9f5b0c52dd29129a0e3d381d7e382a')

build() {
  cd "${srcdir}"/$pkgname-$pkgver

  ./configure --prefix=/usr --disable-inline
  make
}

check() {
  cd "${srcdir}"/$pkgname-$pkgver

  make check
}

package() {
  cd "${srcdir}"/$pkgname-$pkgver

  make DESTDIR="${pkgdir}" install
}
